
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc75285030]Corrections required
0.1 [bookmark: _Toc75285031]Correction to function f_TC_7_1_3_5_1_NR_TestBody
	Function name
	f_TC_7_1_3_5_1_NR_TestBody

	Reason for change
	1) Usage of NR_PDCP_PDUList_Type for the single PDU is leading to mismatch while receiving the PDCP PDU List containing multiple PDUs coming in the same slot
2) At step 7 of the test case , p_NR_PDCP_State.RX_NEXT is set wrongly 


	Summary of change
	1) It is proposed to use the data type NR_PDCP_PDU_Type for the expected individual PDUs 
2) At step 7 of the test case 7.1.3.5.1 , p_NR_PDCP_State.RX_NEXT has to be set with p_NR_PDCP_State.TX_NEXT – 2 instead of having “0” value


	TTCN module
	NR_TC_Common\7_1_3\PDCP_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_7_1_3_5_1_NR_TestBody(NR_PDCP_SS_State_Type p_NR_PDCP_State) runs on NR_BASE_PTC
  {
    var NR_SecurityParams_Type v_NR_Security:= f_NR_Security_Get();
    var SubFrameTiming_Type v_TimingInfo1;
    var SubFrameTiming_Type v_TimingInfo2;
    var SubFrameTiming_Type v_TimingInfok;
    var SubFrameTiming_Type v_TimingUL;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
    var NR_PDCP_SDUList_Type v_SduList;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu1;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu2;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var float v_T_reordering;
    var integer k;
    p_NR_PDCP_State.KUPenc := v_NR_Security.AS_Ciphering.KUPenc;

     //Stop UL Grants
     f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);
    //@siclog "Step 1" siclog@ All PDCP PDUs use same SDU v_IPData
     v_SduList := f_GenerateSDU (p_NR_PDCP_State, {46, 62, 78, 94, 110});

     v_TimingInfo1 := f_NR_GetNextSendOccasion_DL(nr_Cell1);
     v_TimingInfok := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,20);
 
    //@siclog "Step 3" siclog@
    for (k:= 0; k <= 2; k:= k+1) {
      //SS transmits a PDCP Data PDU containing one PDCP SDU
      f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_SduList[k], v_TimingInfok);
      v_TimingInfok := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1, 20*(k+2));
    }
    //@siclog "Step 4" siclog@ Wait Discard timer to expire. @sic R5s190225, R5s190770 sic@
    v_T_reordering := f_NR_SetTimerToleranceMax(nr_Cell1,l2Timer, 500.0);
    v_TimingInfo2 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfok, float2int(v_T_reordering) - 20);

    //@siclog "Step 5" SS transmits a PDCP Data PDU containing one PDCP SDU @sic R5s190225 sic@
    v_TimingInfok := v_TimingInfo2;
    for (k:= 0; k <= 1; k:= k+1) {
      f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1, v_SduList[k+3], v_TimingInfok);
      v_TimingInfok := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo2,20*(k+1));
    }
    
    v_TimingPreviousSlot:= f_NR_GetCurrentTiming(nr_Cell1); //@sic R5-213462 sic@
    //@siclog "Step 6" The SS resumes normal UL grant allocation.
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingInfok, 0); // @sic R5s190590 sic@
    f_NR_ULGrantConfiguration_Start(nr_Cell1, cs_TimingInfo_NR(v_TimingUL));
    //@siclog "Step 7" siclog@ Does UE transmit a PDCP Data PDU # 5 of size 94 bytes? (Note 2)
    //@sic R5-213462 sic@
    v_ExpectedPdu1:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= (v_ExpectedPdu1,v_ExpectedPdu2);
    
    
   alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu1)) -> value v_ReceivedPdu1
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          // receive PDU2
          DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu2)) -> value v_ReceivedPdu2;
          v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
          f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 7-8");
        }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3))-> value v_ReceivedPdu3
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 7-8");
        }
    }

    

  }



After Change:
	  function f_TC_7_1_3_5_1_NR_TestBody(NR_PDCP_SS_State_Type p_NR_PDCP_State) runs on NR_BASE_PTC
  {
    var NR_SecurityParams_Type v_NR_Security:= f_NR_Security_Get();
    var SubFrameTiming_Type v_TimingInfo1;
    var SubFrameTiming_Type v_TimingInfo2;
    var SubFrameTiming_Type v_TimingInfok;
    var SubFrameTiming_Type v_TimingUL;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
    var NR_PDCP_SDUList_Type v_SduList;
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu1;
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu2;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var float v_T_reordering;
    var integer k;
    p_NR_PDCP_State.KUPenc := v_NR_Security.AS_Ciphering.KUPenc;

     //Stop UL Grants
     f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);
    //@siclog "Step 1" siclog@ All PDCP PDUs use same SDU v_IPData
     v_SduList := f_GenerateSDU (p_NR_PDCP_State, {46, 62, 78, 94, 110});

     v_TimingInfo1 := f_NR_GetNextSendOccasion_DL(nr_Cell1);
     v_TimingInfok := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,20);
 
    //@siclog "Step 3" siclog@
    for (k:= 0; k <= 2; k:= k+1) {
      //SS transmits a PDCP Data PDU containing one PDCP SDU
      f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_SduList[k], v_TimingInfok);
      v_TimingInfok := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1, 20*(k+2));
    }
    //@siclog "Step 4" siclog@ Wait Discard timer to expire. @sic R5s190225, R5s190770 sic@
    v_T_reordering := f_NR_SetTimerToleranceMax(nr_Cell1,l2Timer, 500.0);
    v_TimingInfo2 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfok, float2int(v_T_reordering) - 20);

    //@siclog "Step 5" SS transmits a PDCP Data PDU containing one PDCP SDU @sic R5s190225 sic@
    v_TimingInfok := v_TimingInfo2;
    for (k:= 0; k <= 1; k:= k+1) {
      f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1, v_SduList[k+3], v_TimingInfok);
      v_TimingInfok := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo2,20*(k+1));
    }
    
    v_TimingPreviousSlot:= f_NR_GetCurrentTiming(nr_Cell1); //@sic R5-213462 sic@
    //@siclog "Step 6" The SS resumes normal UL grant allocation.
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingInfok, 0); // @sic R5s190590 sic@
    f_NR_ULGrantConfiguration_Start(nr_Cell1, cs_TimingInfo_NR(v_TimingUL));
    //@siclog "Step 7" siclog@ Does UE transmit a PDCP Data PDU # 5 of size 94 bytes? (Note 2)
    //@sic R5-213462 sic@
    p_NR_PDCP_State.RX_NEXT := p_NR_PDCP_State.TX_NEXT - 2; 
    v_ExpectedPdu1:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= {v_ExpectedPdu1,v_ExpectedPdu2}; 
    
    
   alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu1})) -> value v_ReceivedPdu1 
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          // receive PDU2
          DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu2})) -> value v_ReceivedPdu2; 
          v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
          f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 7-8");
        }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3))-> value v_ReceivedPdu3
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 7-8");
        }
    }

    

  }


0.2 [bookmark: _Toc75285032]Correction to function f_TC_7_1_3_5_4_NR_Step1to17
	Function name
	f_TC_7_1_3_5_4_NR_Step1to17

	Reason for change
	In the current TTCN implementation
1) Usage of NR_PDCP_PDUList_Type for the single PDU is leading to mismatch while receiving the PDCP PDU List containing multiple PDUs coming in the same slot
2) At step 14-15, for the PDU List containing multiple PDU in the same slot , the expected combination is populated wrongly


	Summary of change
	1) It is proposed to use the data type NR_PDCP_PDU_Type for the expected individual PDUs 
2) At step 14-15 of the test case 7.1.3.5.4 , the expected combination for multiple PDUs in the same slot should be set with {v_ExpectedPdu1,v_ExpectedPdu2}

	TTCN module
	\ttcn\develop\NR_TC_Common\7_1_3\PDCP_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_7_1_3_5_4_NR_Step1to17(inout NR_PDCP_SS_State_Type p_NR_PDCP_State) runs on NR_BASE_PTC
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var NR_PDCP_SDUList_Type v_SduList := f_GenerateSDU (p_NR_PDCP_State, {64, 164});
    var NR_SecurityParams_Type v_NR_Security:= f_NR_Security_Get();
    var SubFrameTiming_Type v_TimingInfo1;
    var SubFrameTiming_Type v_TimingInfo2;
    var SubFrameTiming_Type v_TimingInfo3;
    var SubFrameTiming_Type v_SubFrameTimingT1;
    var SubFrameTiming_Type v_SubFrameTimingT2;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu1;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu2;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu3;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu4;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu5;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu6;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu4;
    var NR_DRB_COMMON_IND v_ReceivedPdu5;
    var NR_DRB_COMMON_IND v_ReceivedPdu6;
    var float v_T_reordering;
    var TimingInfo_Type v_TimingInfo;
    var integer v_Duration;
    timer t_Wait := 1.0;
    
    p_NR_PDCP_State.KUPenc := v_NR_Security.AS_Ciphering.KUPenc ;
    
    //@siclog "Step 1" siclog@
    //SS transmits a PDCP Data PDU containing one PDCP SDU #131072
    p_NR_PDCP_State.TX_NEXT := 131072; //@sic R5s190227 sic@
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData);
    
    //@siclog "Step 2" siclog@
    //Does the UE transmit a PDCP SDU via the AM RLC entity in the next 1s?
    t_Wait.start;
    alt {
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList(nr_Cell1, p_NR_PDCP_State.DRB_Id)) {
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
      }
      [] t_Wait.timeout{}
    }
    //@siclog "Step 3" siclog@
    //The SS sends the PDCP SDU #1
    p_NR_PDCP_State.TX_NEXT := 1;
    v_TimingInfo1 := f_NR_GetNextSendOccasion_DL(nr_Cell1);
    v_TimingInfo2 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,20);
    v_TimingInfo3 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo2,120);
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo1);
    //@siclog "Step 4" siclog@
    //The SS sends the PDCP SDU #2
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo2);
    //@siclog "Step 5" siclog@ Wait for 100ms (< configured t-Reordering)
    //v_TimingInfo3 - v_TimingInfo2
    //@siclog "Step 6" siclog@ The SS sends the PDCP SDU #0
    p_NR_PDCP_State.TX_NEXT := 0;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo3);
    //@sic R5-213462 sic@
    //@siclog "Step 7-9" siclog@
    //UE transmits the PDCP SDU #0 via the AM RLC entity #0,#1,#2 (Note 4). @sic R5s190227 sic@
    v_ExpectedPdu1:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu4:= (v_ExpectedPdu1, v_ExpectedPdu2);
    v_ExpectedPdu5:= (v_ExpectedPdu2, v_ExpectedPdu3);
    v_ExpectedPdu6:= (v_ExpectedPdu1, v_ExpectedPdu2, v_ExpectedPdu3);
    
    alt {
      // receive PDU1
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu1)) -> value v_ReceivedPdu1
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          alt {
            // receive PDU2
            []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu2)) -> value v_ReceivedPdu2
              {
                v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                v_TimingPreviousSlot:= v_TimingCurrentSlot;
                // receive PDU3
                DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3)) -> value v_ReceivedPdu3;
                v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
                }
             // }
            // receive PDU2 and PDU3
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu5)) -> value v_ReceivedPdu5
              {
                v_TimingCurrentSlot := v_ReceivedPdu5.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
              }
           }
         }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu4)) -> value v_ReceivedPdu4
        {
          v_TimingPreviousSlot := v_ReceivedPdu4.Common.TimingInfo.SubFrame;
            // receive PDU3
            DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3)) -> value v_ReceivedPdu3;
            v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
            f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
         }
      // receive PDU1, PDU2 and PDU3
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu6)) -> value v_ReceivedPdu6
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
        }
    }
    
    //@siclog "Step 10" siclog@
    //The SS sends the PDCP SDU #4
    v_TimingInfo1 := f_NR_GetNextSendOccasion_DL(nr_Cell1); //t-Reordering starts
    v_TimingInfo2 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,100); //Next PDU
    v_TimingInfo3 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,120); //Wait for PDU
    v_SubFrameTimingT1 := v_TimingInfo3;
    p_NR_PDCP_State.TX_NEXT := 4;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo1);
    //@siclog "Step 11" siclog@ Wait for 100ms (< configured t-Reordering)
    //v_TimingInfo2 - v_TimingInfo1
    //@siclog "Step 12" siclog@ The SS sends the PDCP SDU #7
    p_NR_PDCP_State.TX_NEXT := 7;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_SduList[0],v_TimingInfo2);
    //@siclog "Step 13" siclog@ The SS sends the PDCP SDU #3
    p_NR_PDCP_State.TX_NEXT := 3;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo3);
    
    //@siclog "Step 14-15" siclog@ UE transmits a PDCP SDU #3, #4 (Note 5). @sic R5s190227 sic@
    //@sic R5-213462 sic@
    v_ExpectedPdu1:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= (v_ExpectedPdu5,v_ExpectedPdu6);
    
    alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu1)) -> value v_ReceivedPdu1
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          // receive PDU2
          DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu2)) -> value v_ReceivedPdu2;
          v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
          f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 14-15");
        }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3))-> value v_ReceivedPdu3
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 14-15");
        }
    }

    //@siclog "Step 16" siclog@ UE transmits a PDCP SDU #7
    p_NR_PDCP_State.RX_NEXT := 5; //@sic R5-194837 sic@
    v_TimingInfo := f_NR_PDCP_Recv(p_NR_PDCP_State,nr_Cell1,v_SduList[0], cr_RlcBearerRouting_NR(nr_Cell1));
    v_SubFrameTimingT2 := v_TimingInfo.SubFrame;
    // Check 1: Is (T2 - T1) > t-Reordering?
    v_Duration := f_SubFrameTiming_Duration(v_SubFrameTimingT1, v_SubFrameTimingT2);
    v_T_reordering := f_NR_SetTimerToleranceMin(nr_Cell1,l2Timer,300.0);
    if (v_Duration > float2int(v_T_reordering)){
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");
    }
    //@siclog "Step 17" siclog@ The SS sends the PDCP SDU #9
    p_NR_PDCP_State.TX_NEXT := 9;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_SduList[1]);
  }



After Change:
	function f_TC_7_1_3_5_4_NR_Step1to17(inout NR_PDCP_SS_State_Type p_NR_PDCP_State) runs on NR_BASE_PTC
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var NR_PDCP_SDUList_Type v_SduList := f_GenerateSDU (p_NR_PDCP_State, {64, 164});
    var NR_SecurityParams_Type v_NR_Security:= f_NR_Security_Get();
    var SubFrameTiming_Type v_TimingInfo1;
    var SubFrameTiming_Type v_TimingInfo2;
    var SubFrameTiming_Type v_TimingInfo3;
    var SubFrameTiming_Type v_SubFrameTimingT1;
    var SubFrameTiming_Type v_SubFrameTimingT2;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu1;
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu2;
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu3;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu4;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu5;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu6;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu4;
    var NR_DRB_COMMON_IND v_ReceivedPdu5;
    var NR_DRB_COMMON_IND v_ReceivedPdu6;
    var float v_T_reordering;
    var TimingInfo_Type v_TimingInfo;
    var integer v_Duration;
    timer t_Wait := 1.0;
    
    p_NR_PDCP_State.KUPenc := v_NR_Security.AS_Ciphering.KUPenc ;
    
    //@siclog "Step 1" siclog@
    //SS transmits a PDCP Data PDU containing one PDCP SDU #131072
    p_NR_PDCP_State.TX_NEXT := 131072; //@sic R5s190227 sic@
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData);
    
    //@siclog "Step 2" siclog@
    //Does the UE transmit a PDCP SDU via the AM RLC entity in the next 1s?
    t_Wait.start;
    alt {
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList(nr_Cell1, p_NR_PDCP_State.DRB_Id)) {
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
      }
      [] t_Wait.timeout{}
    }
    //@siclog "Step 3" siclog@
    //The SS sends the PDCP SDU #1
    p_NR_PDCP_State.TX_NEXT := 1;
    v_TimingInfo1 := f_NR_GetNextSendOccasion_DL(nr_Cell1);
    v_TimingInfo2 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,20);
    v_TimingInfo3 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo2,120);
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo1);
    //@siclog "Step 4" siclog@
    //The SS sends the PDCP SDU #2
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo2);
    //@siclog "Step 5" siclog@ Wait for 100ms (< configured t-Reordering)
    //v_TimingInfo3 - v_TimingInfo2
    //@siclog "Step 6" siclog@ The SS sends the PDCP SDU #0
    p_NR_PDCP_State.TX_NEXT := 0;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo3);
    //@sic R5-213462 sic@
    //@siclog "Step 7-9" siclog@
    //UE transmits the PDCP SDU #0 via the AM RLC entity #0,#1,#2 (Note 4). @sic R5s190227 sic@
    v_ExpectedPdu1:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu4:= {v_ExpectedPdu1, v_ExpectedPdu2}; 
    v_ExpectedPdu5:= {v_ExpectedPdu2, v_ExpectedPdu3};
    v_ExpectedPdu6:= {v_ExpectedPdu1, v_ExpectedPdu2, v_ExpectedPdu3};
    
    alt {
      // receive PDU1
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu1})) -> value v_ReceivedPdu1         {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          alt {
            // receive PDU2
            []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu2})) -> value v_ReceivedPdu2               {
                v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                v_TimingPreviousSlot:= v_TimingCurrentSlot;
                // receive PDU3
                DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu3})) -> value v_ReceivedPdu3; 
                v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
                }
             // }
            // receive PDU2 and PDU3
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu5)) -> value v_ReceivedPdu5
              {
                v_TimingCurrentSlot := v_ReceivedPdu5.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
              }
           }
         }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu4)) -> value v_ReceivedPdu4
        {
          v_TimingPreviousSlot := v_ReceivedPdu4.Common.TimingInfo.SubFrame;
            // receive PDU3
            DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu3})) -> value v_ReceivedPdu3; 
            v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
            f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
         }
      // receive PDU1, PDU2 and PDU3
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu6)) -> value v_ReceivedPdu6
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7-9");
        }
    }
    
    //@siclog "Step 10" siclog@
    //The SS sends the PDCP SDU #4
    v_TimingInfo1 := f_NR_GetNextSendOccasion_DL(nr_Cell1); //t-Reordering starts
    v_TimingInfo2 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,100); //Next PDU
    v_TimingInfo3 := f_SubFrameTiming_AddMilliSeconds(v_TimingInfo1,120); //Wait for PDU
    v_SubFrameTimingT1 := v_TimingInfo3;
    p_NR_PDCP_State.TX_NEXT := 4;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo1);
    //@siclog "Step 11" siclog@ Wait for 100ms (< configured t-Reordering)
    //v_TimingInfo2 - v_TimingInfo1
    //@siclog "Step 12" siclog@ The SS sends the PDCP SDU #7
    p_NR_PDCP_State.TX_NEXT := 7;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_SduList[0],v_TimingInfo2);
    //@siclog "Step 13" siclog@ The SS sends the PDCP SDU #3
    p_NR_PDCP_State.TX_NEXT := 3;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_IPData,v_TimingInfo3);
    
    //@siclog "Step 14-15" siclog@ UE transmits a PDCP SDU #3, #4 (Note 5). @sic R5s190227 sic@
    //@sic R5-213462 sic@
    v_ExpectedPdu1:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); 
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu4:= {v_ExpectedPdu1,v_ExpectedPdu2}; 
    
    alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu1})) -> value v_ReceivedPdu1 
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          // receive PDU2
          DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu2})) -> value v_ReceivedPdu2; 
          v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
          f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 14-15");
        }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu4))-> value v_ReceivedPdu3 
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 14-15");
        }
    }

    //@siclog "Step 16" siclog@ UE transmits a PDCP SDU #7
    p_NR_PDCP_State.RX_NEXT := 5; //@sic R5-194837 sic@
    v_TimingInfo := f_NR_PDCP_Recv(p_NR_PDCP_State,nr_Cell1,v_SduList[0], cr_RlcBearerRouting_NR(nr_Cell1));
    v_SubFrameTimingT2 := v_TimingInfo.SubFrame;
    // Check 1: Is (T2 - T1) > t-Reordering?
    v_Duration := f_SubFrameTiming_Duration(v_SubFrameTimingT1, v_SubFrameTimingT2);
    v_T_reordering := f_NR_SetTimerToleranceMin(nr_Cell1,l2Timer,300.0);
    if (v_Duration > float2int(v_T_reordering)){
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");
    }
    //@siclog "Step 17" siclog@ The SS sends the PDCP SDU #9
    p_NR_PDCP_State.TX_NEXT := 9;
    f_NR_PDCP_Send(p_NR_PDCP_State,nr_Cell1,v_SduList[1]);
  }
    



